

Chapitre 1 :

Le TABLEUR :
AUTOMATISATION DE LA
RESOLUTION DES PROBLEMES
DE GESTION

Chapitre 2 :

Le TABLEUR : FONCTIONS DE
RECHERCHE, CONTRÔLE ET
ERGONOMIE DES FEUILLES

1	Chapitre 1 : Le TABLEUR : AUTOMATISATION DE LA RESOLUTION DES PROBLEMES DE GESTION	3
1.1	LE FONCTIONNEMENT GENERAL DU TABLEUR	4
1.1.1	Une première approche du tableur.....	4
1.1.2	Les composantes d'une feuille de calcul Excel	5
1.2	LES FORMULES DE CALCUL	6
1.3	LES MODES DE REFERENCE AUX CELLULES	7
1.3.1	Les références absolues, relatives, ou mixtes	7
1.3.2	La notion de nom d'une cellule ou d'une plage de cellules.	8
1.4	LES FONCTIONS DE BASE	9
1.4.1	Les principes généraux	9
1.4.2	Les fonctions les plus courantes.....	9
1.4.3	Les fonctions logiques	10
1.4.4	Fonction « ET » et « OU ».....	11
1.4.5	Autres fonctions logiques.....	12
1.4.6	Les fonctions portant sur les dates.....	13
1.5	SYNTHESE.....	15

2	Chapitre 2 : Le TABLEUR : FONCTIONS DE RECHERCHE, CONTRÔLE ET ERGONOMIE DES FEUILLES	16
2.1	FONCTIONS DE RECHERCHE	17
2.1.1	Fonction RECHERCHE (forme générale)	17
2.2	FONCTIONS RECHERCHEV ET RECHERCHEH	19
2.2.1	Fonction RECHERCHEV	19
2.3	FONCTIONS INDEX ET EQUIV	22
2.3.1	Fonction INDEX	22
2.3.2	Fonction EQUIV	23
2.4	IMBRICATION DES FONCTIONS	25
2.5	FONCTIONS FINANCIERES	26
2.6	FONCTIONS STATISTIQUES	27
2.7	ERGONOMIE D'UNE FEUILLE DE CALCUL	28
2.7.1	Fonction d'information EST	28
2.7.2	Fonction SIERREUR	29
2.7.3	Fonction CONCAT	30
2.7.4	Formats de cellules	31
2.7.5	Validation des données	32
2.8	SYNTHESE	33

1 Chapitre 1 : Le TABLEUR : AUTOMATISATION DE LA RESOLUTION DES PROBLEMES DE GESTION

COMPETENCES ATTENDUES	SAVOIRS ASSOCIES
Automatiser des calculs en écrivant des formules Concevoir et modifier la structure d'un modèle de feuille de calcul	Découverte du tableur : Le classeur, la feuille de calcul, la cellule et les types de données Formules utilisant des fonctions logiques, de calculs appliqués aux nombres et aux dates

1.1 Le fonctionnement général du tableur

Définition :

Le Tableur est un progiciel-outil, donc un ensemble cohérent de programmes qu'un utilisateur peut activer selon ses besoins en s'appuyant sur une documentation.

1985 Sort la première version d'Excel

Conçu initialement pour effectuer des calculs complexes, le tableur offre de nombreuses fonctionnalités complémentaires utiles au gestionnaire.

1.1.1 Une première approche du tableur

Le tableur gère des **feuilles de calcul** (des tableaux à deux dimensions) composées de **cellules** (intersections d'une ligne et d'une colonne) où sont rangées des informations. Une cellule peut contenir :

-  Du texte,
-  Des valeurs numériques
-  Des formules

Les feuilles de calcul sont regroupées dans des classeurs, chaque classeur contenant des feuilles qui peuvent être liées entre elles.

Le contenu d'une cellule est limité à 32 767 caractères maxi y compris les caractères génériques (* / : §...) et les espaces.

DU TEXTE Alignées à gauche Référencées entre "" dans les formules de calcul Précédé d'un ' pour imposer une donnée en tant que texte Pour compter du texte, utiliser la fonction de calcul =NBVAL	Budget 2011 2011 budget 1,000.00 29/02/2011 100 km
DES VALEURS NUMERIQUES Alignées à droite Ils sont positifs ou négatifs Si négatifs, il faut taper un – avant le nombre ou le mettre entre parenthèses () La date est une donnée nombre, à condition de respecter la forme de la saisie JJ/MM/AA Le calendrier d'Excel par défaut commence au 01/01/1900	125 125,52 -110 28/02/2011 09 :00 :00 19,6%
DES FORMULES Alignées à droite Elle est toujours précédée d'un symbole = Ne pas oublier les parenthèses pour rendre crédible une formule de calcul Deux façons de calculer : Avec Les opérateurs arithmétiques : + Addition - Soustraction / Division * Multiplication ^ Puissance Avec les Formules de calculs =somme() =moyenne() ... etc	=F3+F4 =(10+10)/2 =somme(A1 :A600) somme de A1 à A600 =somme(A1 ;A600) somme de A1 et A600 UNION DE PLAGES : la somme de l'union des deux plages, il suffit d'utiliser la fonction SOMME et de l'appliquer aux deux plages séparées par un point virgule : =somme(B3 :C10 ;C8 :D13) INTERSECTION DE PLAGES : la somme de l'intersection des deux plages il faut donc écrire la somme sur les deux plages séparées par un espace soit =somme(B3 :C10 C8 :D13)

1.1.2 Les composantes d'une feuille de calcul Excel

Une feuille de calcul présente principalement **trois zones** :

- Un **quadrillage** représentant les cellules de la feuille, où sont placés données et les calculs ;
- Une **ligne d'édition**, sur laquelle apparaît le détail du contenu de la cellule active à un moment donné (y figurent également les cadres permettant de nommer la cellule ou d'y définir une fonction standard de calcul) ;
- Un **bandeau** (appelé RUBAN) regroupant les menus donnant accès aux fonctionnalités du tableur




Figure 1 Principaux affichages d'une feuille Excel




Figure 2 Principaux affichages d'une feuille Excel

1.2 Les formules de calcul

Tout traitement contenu dans la feuille de calcul nécessite de sélectionner la ou les cellules concernées puis de choisir l'action à effectuer. Une cellule est repérée par ses coordonnées (colonne et ligne). Une **plage de cellules** est délimitée par les coordonnées de ses cellules extrêmes.

L'expression de **la formule de calcul** placée dans une cellule commence toujours par le signe =.

Elle fait généralement appel à des opérateurs arithmétiques :

- + Addition
- Soustraction
- / Division
- * Multiplication
- ^ Puissance

Les formules de calcul peuvent faire référence à d'autres cellule de la feuille (sauf quand elles servent à initialiser une valeur, par exemple en y recopiant le contenu d'une autre cellule). Elles peuvent intégrer des fonctions qui, elles-mêmes, comprennent du texte et des symboles spéciaux ou de comparaison :

- > Supérieur strictement
- >= Supérieur ou égal
- < Inférieur strictement
- <= Inférieur ou égal
- <> Différent
- = Egal

Exemple :

=A2*5, effectue la multiplication de la valeur de la cellule A2 par 5,

= C8, correspond à la reprise de la valeur placée en C8

=somme(C10 :C12) fait référence à une fonction qui calcule C10+C11+C12 (la somme des cellules comprises entre C10 et C12, donc la plage C10 :C12)

1.3 Les modes de référence aux cellules

1.3.1 Les références absolues, relatives, ou mixtes

Le contenu d'une cellule peut être recopié dans une ou plusieurs d'entre elles, notamment par les commandes COPIER/COLLER.

Les textes et les valeurs sont copiés sans modification.

En revanche, lors de la copie d'une formule, les références à des cellules contenues dans cette dernière sont adaptées en fonction du décalage entre la cellule d'origine et la ou les cellules de destination (translation de références).

	A	B	C
1	10	20	=A1+B1
2	15	16	=A2+B2

La formule en C1 fait référence de façon relative à A1 et B1

Elle est copiée à partir de C1 et collée en C2

Elle s'adapte par translation et C2 fera référence à A2 et B2

Pour éviter toute modification des formules, il faut faire précéder l'indice de ligne et/ou de colonne des cellules concernées par le symbole de référence absolue \$

Dans le cas précédent, une formule =A\$1+B\$2, saisie en C1 resterait =A\$1+B\$2 en C2.

FORMULE EN A1 COPIEE EN C3				A1=		C3=	TYPE DE REFERENCE																
<table border="1"> <thead> <tr> <th></th> <th>A</th> <th>B</th> <th>C</th> </tr> </thead> <tbody> <tr> <td>1</td> <td>=B1+\$B1+B\$1+\$B\$1</td> <td></td> <td></td> </tr> <tr> <td>2</td> <td></td> <td></td> <td></td> </tr> <tr> <td>3</td> <td></td> <td></td> <td>=D3+\$B3+D\$1+\$B\$1</td> </tr> </tbody> </table>					A	B	C	1	=B1+\$B1+B\$1+\$B\$1			2				3			=D3+\$B3+D\$1+\$B\$1	B1	→	D3	Relative : Ligne et colonne se décalent
					A	B	C																
				1	=B1+\$B1+B\$1+\$B\$1																		
				2																			
3			=D3+\$B3+D\$1+\$B\$1																				
\$B1	→	\$B3	Colonne absolue, ligne relative																				
B\$1	→	D\$1	Colonne relative, ligne absolue																				
\$B\$1	→	\$B\$	Colonne absolue, ligne absolue																				

La feuille ci-après analyse les ventes en calculant leur montant absolu et relatif par mois.

	A	B	C	D
1	Tarif 12 galettes	5,00 €		
2	Quantité vendues (en douzaine)		%	CA
3	Semaine 1	40	12,50%	200,00 €
4	Semaine 2	80	25,00%	400,00 €
5	Semaine 3	140	43,75%	700,00 €
6	Semaine 4	60	18,75%	300,00 €
7	...			
8	Total mois	320		

La formule implantée en C3 : =B3/\$B8, recopiée jusqu'en C6, soit en C4 : =B4/B\$8

La formule implantée en D3 : =B3*B\$1, recopiée jusqu'en D6, soit en D4 : =B4*B\$1

Il est inutile d'utiliser une référence absolue pour la colonne puisque l'on recopie les formules en colonnes ; une référence mixte suffit comme B\$1.

1.3.2 La notion de nom d'une cellule ou d'une plage de cellules.

Une cellule ou une plage de cellules peut être nommée pour y faire référence sans craindre un décalage intempestif, tout en améliorant la lisibilité des formules concernées.

Le recours à des noms permet également de retrouver plus facilement les erreurs.

Pour créer un nom, il est possible d'utiliser le gestionnaire de noms du menu FORMULES ou de l'implanter directement dans la zone de nom (cf figure 1 Titre : 1.1.2).




Figure 3 Création d'un nom

La définition d'une zone nommée attribue le caractère absolu à toutes les références de la zone.

Ce nom est alors utilisable à la place de la référence à la cellule ou à la plage de cellules.

Exemple :

Dans la feuille qui précède les sommes des cellules B8 et D8 peuvent s'écrire : =somme(B3 :B6) et =somme(D3 :D6). La formule en D8 peut être obtenue par copie de B8, car il y a translation.

Si la plage B3 :B6 était nommée VENTES, la formule en B8 pourrait s'écrire : =somme(VENTES). Dans ce cas, elle ne pourrait plus être recopiée en D8.

On peut également nommer une valeur, quand on veut y faire référence, sans l'implanter dans les formules de calcul, ce qui est particulièrement utile pour des paramètres comme le taux de la TVA.

Exemple :

Nom	Valeur	Fait référence à	Étendue	Commentaire
TARIF	5,00 €	=Feuil3!\$B\$1	Classeur	
tva	20%	=Feuil4!\$A\$1	Classeur	

Conseil :

Utiliser des références absolues, relatives ou mixtes ou avoir recours à des noms de façon performante et cohérente, tout en anticipant les copies éventuelles, est un impératif professionnel.

1.4 Les fonctions de base

1.4.1 Les principes généraux

Des fonctions prédéfinies sont disponibles dans le tableur pour effectuer des traitements standardisés, des calculs statistiques ou autres opérations mathématiques.

La forme générale d'une fonction est :

=Nom_fonction(argument1 ;argument2 ;...)

Le nombre d'arguments dépend de la fonction et certains peuvent être optionnels. L'argument peut être une référence à une cellule ou à une plage de cellules, à une valeur ou à une autre fonction ou expression.

Exemple :

La fonction somme évite de devoir exprimer les additions en détail.

De la même façon, la fonction =moyenne(A1 :A10) calcule la moyenne des cellules A1, A2, ... A10.

=aujourd'hui() est une fonction sans argument, qui fournit la date du jour.

1.4.2 Les fonctions les plus courantes

Sont ici présentées les fonctions du tableur les plus fréquemment utilisées pour construire des feuilles de calcul en gestion.

Bien d'autres sont accessibles et documentées dans l'aide du tableur.

FONCTION	EXEMPLE	DESCRIPTION
SOMME	=SOMME(B10 :B15)	Effectue la sommes des valeurs numériques comprises dans la plage B10 :B15inclus (les cellules vides ou contenant du texte sont ignorées)
MOYENNE	=MOYENNE(B10 :C12)	Effectue la moyenne des valeurs numériques situées entre B10et C12 inclus (les cellules vides ou contenant du texte sont ignorées)
MIN	=MIN(B10 :B15)	Affiche la valeur la moins élevée de la plage citée
MAX	=MAX(B10 :B15)	Affiche la valeur la plus élevée de la plage citée
NB	=NB(B10 :B15)	Détermine le nombre de cellules de la plage citée contenant des nombres uniquement
NBVAL	NBVAL(B10 :B15)	Détermine le nombre de cellules de la plage citée contenant tout type de données (nombres, texte)
ESTVIDE	ESTVIDE(B6)	Renvoie vrai si la cellule B6 est vide et faux dans le cas contraire (cette fonction permet

		notamment d'effectuer un calcul à la condition qu'une donnée existe dans une cellule, ce qui évite des messages d'erreur)
ARRONDI	=ARRONDI(B6 ;3) =ARRONDI(MOYENNE(B10 :C12) ;1)	Arrondi la valeur de la cellule B6 à trois chiffres après la virgule Arrondit à une décimale la moyenne calculée
Autres fonctions d'arrondis : ARROND.SUP, ARRONDI .INF, qui arrondissent une valeur à la valeur supérieure ou inférieure.		

Les fonctions de calcul (SOMME, MOYENNE ...) sont d'un usage plus souple que les formules explicites car elles ignorent les cellules qui ne contiennent pas de valeur numérique.

1.4.3 Les fonctions logiques

La fonction SI est la fonction la plus utilisée dans les feuilles de calcul. Elle permet de réaliser une alternative et s'écrit ainsi :

=SI(expression de condition ;action ou valeur si vrai ; action ou valeur si faux)

Les conditions et actions peuvent porter sur des valeurs, des références de cellules, des textes (un texte est pris dans une cellule ou mentionné entre guillemets).

Exemple :

=SI(A10>300 ;A10*B1 ;A10*B2)

=SI(C35=G32 ;D5*F8 ; « Calcul impossible »)

Dans ce cas, le texte « Calcul impossible » apparaît dans la cellule qui contient le SI quand la valeur en C35 diffère de celle qui est en G32.

Plusieurs fonctions SI peuvent être imbriquées pour tenir compte de combinaisons complexes de conditions (64 max).

Exemple :

Les commissions versées aux commerciaux de l'Entreprise LOCATOUT sont proportionnelles au chiffre d'affaires. Ce calcul se traduit par le tableau suivant :

	A	B	C	D	E	F	G
1	Données relatives aux tranches de CA						Taux de commission
2	Tranche 1	0	>	CA	<=	15 000,00 €	2,00%
3	Tranche 2	15 000,00 €	>	CA	<=	30 000,00 €	4,00%
4	Tranche 3	30 000,00 €	>	CA			5,50%
5	NOM	CA	TAUX	COMMISSION			
6	Herbert	45 000,00 €	5,50%	2 475,00 €			
7	Le Squer	12 000,00 €	2,00%	240,00 €			
8	Hernandes	20 000,00 €	4,00%	800,00 €			

Les commissions de la plage C6 :C8 sont déterminées selon l'algorithme suivant :



Formule en C6 =SI(B6>\$B\$4 ;\$G\$4 ;SI(B6>\$B\$3 ;\$G\$3 ;\$G\$2))

Elle est ensuite recopiée vers C8. Les constantes (paramètres) sont placées dans les zones de données B2 :B4 et F2 :G4. Un changement de ces constantes n'implique donc pas de modifier la formule.

La fonction ESTVIDE est fréquemment associée à une alternative « fonction SI » pour effectuer un calcul à condition qu'une donnée existe dans une cellule, ce qui évite des messages d'erreur.

1.4.4 Fonction « ET » et « OU »

Les fonctions logiques ET et OU sont souvent utilisées pour exprimer des conditions complexes dans un SI. Leur forme générale est la suivante :

ET (Condition1 ; Condition2 ; ... Condition N)

OU (Condition1 ; Condition2 ; ... Condition N)

Elles permettent de combiner plusieurs conditions et renvoient une valeur VRAI ou FAUX selon que la combinaison est satisfaite ou non.

Exemple :

=SI(OU(B30= »Payé » ;H56<10) ; »Ne rien faire » ; »relance »

Cette formule affiche « Ne rien faire si la cellule B30 contient la valeur « Payé » ou si la cellule H56 contient une valeur inférieure à 10. A défaut, elle affiche le texte « Relance ».

1.4.5 Autres fonctions logiques

D'autres fonctions logiques permettent d'effectuer des calculs sous conditions.

SOMME.SI	Effectue la somme des cellules d'une plage répondant à une certaine condition	
	=SOMME.SI(C4 :C8 ; « <15 »)	Effectue la somme des valeurs inférieures à 15 de la plage C4 :C8 La condition est obligatoirement placée être guillemets)
	=SOMME.SI(C4 :C8 ;20 ;D4 :D8)	Dans la plage D4 :D8 on n'additionnera que les valeurs pour lesquelles les cellules correspondantes de la plage C4 :C8 contiendront la valeur 20
On dispose de même des fonctions MOYENNE.SI, NB.SI		
SOMME.SI.ENS	=SOMME.SI.ENS(C2 :C8 ;A2 :A8 ; « >10 » ;A2 :A8 ; « <15 »)	Dans la plage C2 :C8 on n'additionnera que les valeurs pour lesquelles les cellules correspondantes de la plage A2 :A8 contiennent des valeurs supérieures à 10 et inférieures à 15
MOYENNE.SI.ENS	=MOYENNE.SI.ENS(C2 :C8 ;A2 :A8 ; « >10 » ;A2 :A8 ; « <15 » ;B2 :B8 ; « France »)	Dans la plage C2 :C8 on ne fera la moyenne que des valeurs pour lesquelles les cellules correspondantes de la plage A2 :A8 contiennent des valeurs supérieures à 10 et inférieures à 15 et pour lesquelles les cellules correspondantes de la plage B2 :B8 contiennent la valeur France
NB.SI.ENS	=NB.SI.ENS(A2 :A8 ; « >10 » ;A2 :A8 ; « <15 » ;D2 :D8 ; « Nouveau client »)	Dans la plage A2 :A8 on ne dénumbrera que les valeurs pour lesquelles les cellules de la plage A2 :A8 contiennent des valeurs supérieures à 10 et inférieures à 15 et pour lesquelles les cellules correspondantes de la plage D2 :D8 contiennent la valeur « nouveau client » (Il y a un argument de moins dans cette fonction)

	A	B	C	D	E	L	M	N	O	P	Q
1	DCG RESULTATS AUX EXAMENS										
2	ANNEE	2016							Total Coeff		
3	Coefficient	1	1	1	1	1,5	1	1	15		
4		UE1	UE2	UE3		UE11	UE12	UE13	Moy	NB UE <6	examen
5	ETU 1	8	12	11,75		15	12,5	11,75	10,85		Admis(e)
6	ETU 2	8	8,75	10,25		14,5	5	11,25	11,08	1	
7	.../...										
8	ETU 40	10	10	10		14,5	11,25	8,25	10,73		Admis(e)
9	% de réussite par UE	80%	70%	98%		88%	88%	83%			
10											
11	Moyenne par UE	11,3	9,7	13		12,7	12	11,5	11,8		
12	Note minimum par UE	5,5	38	8,5		4	5	5			
13	Note maximum par UE	14,5	13,5	18,8		18,8	19	18,8			
14	Nb notes <6	2	6	0		2	1	3			
15	Nb notes <=6 et <10	6	6	1		3	4	4			
16	Nb notes <=10	32	28	39		35	35	33			

La plage B3 :N3 a été nommée « coefficient ». Les formules ont été recopiées vers la droite ou vers le bas en fonction des besoins et ont donc été conçues comme telles. On relève notamment les formules suivantes :

ADRESSE	FORMULE DE CALCUL
O3	=SOMME(COEFFICIENT)
O5	=ARRONDI(SOMME(coefficient ;B5 :N5)/O\$3 ;2) Formule recopiée vers O44
B45	=ARRONDI(B52/NB(B5 :B44) ;2)
B47	=ARRONDI(MOYENNE(B5/B44) ;1)
B48	=ARRONDI(MIN(B5 :B44) ;1)
B49	=ARRONDI(MAX(B5 :B44) ;1)
B50	=NB.SI(B5 :B44 ; « <6 »)
B51	=NB.SI.ENS(B5 :B44 ; « >=6 » ;B5 :B44 ; « <10 »)

1.4.6 Les fonctions portant sur les dates

Toute date est mémorisée par Excel, grâce au nombre de jours écoulés depuis le 01/01/1900. Une différence de dates fournit donc un nombre de jours.

La fonction =DATE(année ;mois ;jour) renvoie le nombre de jours définissant la date placée en argument de la fonction.

Exemple :

La fonction =DATE(2020 ;1 ;20) renvoie la valeur 43850 au format nombre, soit le 20/01/2020 au format date.

D'autres fonctions courantes portent sur les dates :


La fonction ANNEE(date) fournit l'année de la date désignée (elle peut être soit saisie en argument entre guillemets, soit contenue dans une cellule placée en argument)

De même, les fonctions MOIS, JOUR ou HEURE renvoient respectivement le mois, le jour ou l'heure d'une date ;

La fonction AUJOURDHUI() renvoie la date du jour connue de l'ordinateur au format date ;

La fonction JOURS360 permet de calculer le nombre de jours entre deux dates sur la base d'une année de 360 jours (elle est très utile en comptabilité pour calculer un prorata temporis)

Exemple :



	A	B	C	D	E
1	12/11/2020	2020			
2					

A1 affiche 12/11/2020, =ANNEE(A1) renvoie la valeur 2020

	A	B	C	
1	15/1/2020	31/3/2020	76	C1 : =JOURS360(A1 ;B1)

1.5 SYNTHÈSE

GENERALITES

Le tableur gère des feuilles de calcul, éventuellement regroupées dans des classeurs, composées de cellules qui peuvent comprendre du texte, des images, des valeurs numériques ou des formules de calcul.

Les formules de calcul portent sur les cellules repérées par leurs coordonnées, elles commencent toujours par le signe égal, utilisent des opérateurs et intègrent souvent des fonctions.

Ces formules peuvent faire référence aux cellules de façon absolue, relative ou mixte, les cellules ou les plages de cellules pouvant également être nommées.

FONCTIONS COURANTES

Les fonctions ont un nom, s'écrivent en majuscules et comprennent un nombre variable d'arguments, les plus courantes sont les suivantes :

FORMAT	SOMME	ESTVIDE
	MOYENNE	ARRONDI
	MAX	ARRONDI.SUP
	MIN	ARRONDI.INF
	NB	NBVAL
	NB.SI	NB.SI.ENS
	SOMME.SI	SOMME.SI.ENS
	SOMMEPROD	

FONCTIONS LOGIQUES COURANTES

Les fonctions SI offrent une alternative comme suit :

SI(Expression de condition ;action ou valeur si VRAI ;action ou valeur si FAUX)

La fonction ESVIDE est fréquemment associée à la fonction SI pour éviter des messages d'erreur

La fonction SI peut porter sur plusieurs conditions combinées associées par le recours aux fonctions ET et OU.

2 Chapitre 2 : Le TABLEUR : FONCTIONS DE RECHERCHE, CONTRÔLE ET ERGONOMIE DES FEUILLES

COMPETENCES ATTENDUES	SAVOIRS ASSOCIES
<p>Automatiser des calculs en écrivant des formules</p> <p>Concevoir et modifier la structure d'un modèle de feuille de calcul</p> <p>Contrôler l'adéquation entre le contexte d'un problème de gestion et le modèle créé</p> <p>Mettre en place l'ergonomie d'une feuille de calcul</p>	<p>Découverte du tableur : Le classeur, la feuille de calcul, la cellule et les types de données</p> <p>Éléments d'ergonomie (formatage des cellules, gestion de l'affichage, validation des données, formatage conditionnel et gestion des erreurs)</p>

2.1 Fonctions de recherche

Définition :

Les fonctions de recherche permettent de trouver la valeur d'une liste associée à une autre valeur placée dans une autre liste.

Exemple :

Une catégorie (codée A, B ou C) est affectée à chaque client. A chacune des catégories de la liste A- B – C correspond un taux de remise (dans l'ordre 0, 2 ou 4%). A partir du code de catégorie d'un client, la recherche dans la liste des codes permet de trouver le taux de remise à appliquer.

Les listes forment une table implantée au préalable dans une feuille de calcul active ou dans une autre feuille de calcul.

Les fonctions de recherche nécessitent, en général, que la première liste dans laquelle on cherche une valeur pour lui trouver une correspondance soit triée.

2.1.1 Fonction RECHERCHE (forme générale)

PRINCIPE :

La forme générale de la fonction RECHERCHE est :

=RECHERCHE(valeur à chercher ;colonne ou ligne de recherche ;colonne ou ligne de résultats)

La fonction RECHERCHE recherche une valeur dans une colonne (ou une ligne) d'un tableau et renvoie une valeur située à la même position dans la colonne (ou la ligne) qui contient les résultats.

Exemple :

Dans le tableau qui suit, le taux de remise du client « Optica » placé en C8 peut être déterminé directement, en fonction de la catégorie placée en C6, par la formule =RECHERCHE(C6 ;E3 :E5 ;F3 :F5)

	A	B	C	D	E	F
1	REMISES CLIENTS					
2	CODE CLIENT	NOM	CATEGORIE		CATEGORIE	REMISE
3	124	LE GALL	A		A	0%
4	132	HUMEAU	A		B	2%
5	329	SA LEGRAND	C		C	4%
6	543	OPTICA	B			
7						
8	REMISE OPTICA		2%			

Si la table CATEGORIES/REMISES était placée dans la feuille de calcul externe nommée CAT ci-après, la formule de recherche précédente deviendrait :

=RECHERCHE(C6 ;CAT !A2 :A4 ;CAT !B2 :B4)

	A	B
1	CATEGORIE	REMISE
2	A	0%
3	B	2%
4	C	4%

La feuille qui suit montre comment il est possible de calculer directement le montant de la remise en fonction du code du client placé en B10, à l'aide de **recherches imbriquées**.

B11								
=RECHERCHE(RECHERCHE(B10;A3:A6;C3:C6);E3:E5;F3:F5)*B9								
	A	B	C	D	E	F	G	H
1	REMISES CLIENTS							
2	CODE CLIENT	NOM	CATEGORIE		CATEGORIE	REMISE		
3	124	LE GALL	A		A	0%		
4	132	HUMEAU	A		B	2%		
5	329	SA LEGRAND	C		C	4%		
6	543	OPTICA	B					
7								
8	exemple calcul de remise							
9	VENTE	23166						
10	CLIENT	329						
11	REMISE	926,64						

En B11 on trouve : =RECHERCHE(RECHERCHE(B10;A3:A6;C3:C6);E3:E5;F3:F5)*B9

LISTES DISJOINTES

La fonction RECHERCHE permet si nécessaire, d'exploiter des données de listes ayant des directions différentes, horizontales ou verticales. Il est également possible d'effectuer des recherches par tranches puisque le tableur retient, par défaut, la catégorie correspondant au minimum de la tranche.

2.2 Fonctions RECHERCHEV et RECHERCHEH

La fonction RECHERCHEV joue le même rôle que la fonction RECHERCHE, mais elle effectue une recherche verticale, par colonnes. De même, la fonction RECHERCHEH recherche horizontalement par lignes.

LISIBILITE RENFORCEE

Les fonctions RECHERCHEV et RECHERCHEH, quand elles peuvent être mobilisées, permettent une expression plus rapide et plus lisible des formules, c'est pourquoi elles sont, le plus souvent préférées à la forme générale.

2.2.1 Fonction RECHERCHEV

La fonction RECHERCHEV permet de rechercher verticalement par l'index de la première colonne, à partir d'une valeur spécifiée, dans une table, la valeur correspondante dans une autre colonne indiquée et ce de façon exacte ou approximative.

Sa forme générale est la suivante :

=RECHERCHEV(valeur_recherchée ;table_matrice ;numéro_index_colonne,valeur_proche ;(vrai/faux)

Valeur que l'on va rechercher dans la première colonne de la table	Table de données dans laquelle on effectue la recherche	Numéro de la colonne dans laquelle se trouve la valeur que l'on recherche	Permet de faire une recherche exacte ou approchée
--	---	---	---

La table_matrice peut être nommée ou on peut y faire référence par une plage.

Le quatrième argument est « valeur_proche » (VRAI/FAUX) :

S'il est VRAI et que la valeur cherchée ne se trouve pas dans la liste ou que la première colonne n'est pas triée par ordre croissant (arrêt de la recherche dès qu'on atteint une valeur supérieure) alors c'est la première valeur immédiatement inférieure à la valeur recherchée qui sera retenue. Il permet éventuellement une recherche par tranche ce qui peut s'avérer utile en gestion.

S'il est FAUX, une recherche exacte sera effectuée. Si la valeur cherchée ne se trouve pas dans la liste ou que la première colonne n'est pas triée, la recherche ne pourra être effectuée et la fonction renverra le message d'erreur #N/A.

Exemple :

On souhaite retrouver le nom d'un salarié dans une table en fonction de son matricule.

B3		=RECHERCHEV(B2;\$D\$3:\$F\$5;2;FAUX)				
	A	B	C	D	E	F
1				table_matrice		
2	Matricule	HE-512		Matricule	Nom	Prénom
3	Nom	THOLAS		HE-511	HERBERT	L2A
4				HE-512	THOLAS	BERTRAND
5				FG-513	GARCIA	FLORE
6			Sens de la recherche		Colonne 2 : données correspondantes	
7						
8						

B3 : =RECHERCHEV(B2;\$D\$3:\$F\$5;2;FAUX)

Premier argument	Deuxième argument	Troisième argument	Quatrième argument
B2	\$D\$3:\$F\$5	2	FAUX
La valeur à chercher se trouve dans la cellule B2, soit ici HE-512	Table contenant les données	Index (rang) de la colonne où on trouve les valeurs de la donnée cherchée, ici 2 il s'agit donc des noms	La recherche se fera de façon exacte ce qui est obligatoire ici car les matricules ne sont pas classés par ordre alphabétique

Exemple :

Le taux de ristourne de l'Entreprise Gallo, client de l'entreprise Igloo, est obtenu comme suit :

B3		=RECHERCHEH(B2;E3:J4;2;VRAI)								
	A	B	C	D	E	F	G	H	I	J
1	Nom client	GALLO		Pourcentages de ristournes octroyées selon le montant total des chiffres d'affaires (intervalles de ces montants en colonnes ci-dessous)						
2	Chiffre d'affaires	125 000,00 €	Sens de la recherche							
3	Taux de ristourne	0,92%		Seuils	0	50 000	10 000	200 000	400 000	600 000
4	Montant de ristourne	1 150,00 €		Taux	0,75%	0,80%	0,92%	0,96%	0,97%	0,99%
5					Ligne 2 : données correspondantes					

Le quatrième argument doit être omis car il s'agit d'une recherche par tranches.

CONSEIL

Lorsque vous utilisez la fonction RECHERCHE la mobilisation des arguments est identique. Le troisième argument indique l'index de la ligne et non plus de la colonne à partir de laquelle la donnée correspondante est renvoyée.

METHODE

Principes fondamentaux des fonctions de recherche

- Veillez à ce que le premier argument recherché soit bien situé dans la première ligne ou première colonne de la matrice indiquée au sein du deuxième argument.
- Dans certains contextes, il est nécessaire d'omettre le quatrième argument ou de mettre VRAI : travaux sur des dates, des intervalles (âge, tranches d'imposition ...)
- Dans d'autres contextes, il est indispensable d'utiliser le quatrième argument : Tri non croissant de la matrice du deuxième argument ou vérification d'une saisie effectuée par ailleurs.

2.3 Fonctions INDEX et EQUIV

Quand la recherche ne consiste ni à rechercher en fonction de l'index de la première colonne, ni en fonction de celui de la première ligne mais à l'intersection d'une ligne et d'une colonne, la fonction INDEX doit être mobilisée. Elle sélectionne une valeur dans un tableau à partir des index de numéros de ligne et de colonne, qui peuvent être paramétrés à l'aide des fonction EQUIV ou RECHERCHE.

2.3.1 Fonction INDEX

La forme générale de la fonction INDEX est la suivante :

=INDEX(Tableau ; no_ligne ; no_colonne)

Premier argument : Matrice (tableau) où l'on va rechercher une valeur	Deuxième argument : N° de la ligne	Troisième argument : N° de la colonne
Les argument 2 et 3 peuvent être retrouvés avec la fonction EQUIV		

Les arguments no_ligne et no_colonne doivent pointer sur une cellule appartenant à l'argument tableau. A défaut, la fonction INDEX renvoie la valeur d'erreur #REF !

Exemple :

Le prix de l'article Jupe chez Galopin peut être recherchée par une fonction INDEX :

B2	=INDEX(\$E\$3:\$F\$5;2;1)					
	A	B	C	D	E	F
1	Article	Jupe		Données		
2	Prix	45		Produit	Prix	Qté
3				Chemise	30	200
4			Ligne 2	Jupe	45	300
5				Pantalon	55	150
6					Colonne 1	

B2 : =INDEX(\$E\$3:\$F\$5;2;1)

Premier argument	Deuxième argument	Troisième argument
\$E\$3:\$F\$5	2	1

Table contenant les données	Numéro de la ligne de la table où se trouve les valeurs de la donnée cherchée, ici 2	Numéro de la colonne de la table où on trouve les valeurs de la donnée cherchée, ici 1
-----------------------------	--	--

2.3.2 Fonction EQUIV

La fonction EQUIV permet d'automatiser la détermination de la ligne ou de la colonne (deuxième et Troisième arguments) de la fonction INDEX.

EQUIV recherche une valeur spécifique dans une plage de cellules puis renvoie sa position relative (numéro de ligne ou de colonne) dans la plage.

La forme générale de la fonction EQUIV est la suivante :

=EQUIV(Valeur cherchée ; matrice recherche; type)

Valeur dont on recherche la position dans la liste	Liste dans laquelle se trouve la valeur	Argument facultatif dont le fonctionnement s'apparente à celui du quatrième argument de la fonction recherche
--	---	---

Le troisième argument, type est facultatif et peut prendre les valeurs suivantes :

- 0 Recherche une correspondance exacte
- 1 (ou omis) permet de chercher une valeur strictement inférieure ou égale à la valeur cherchée
- -1 permet de chercher une valeur strictement supérieure à la valeur cherchée
- 1 ou omis et -1 permettent des recherches par tranches.

Exemple :

La jupe de Galopin devient :

B2 X ✓ fx =INDEX(\$E\$3:\$F\$5;EQUIV(B1;\$D\$3:\$D\$5;1);EQUIV(A2;\$E\$2:\$F\$2;1))

	A	B	C	D	E	F	G	H	I
1	Article	Jupe		Données					
2	Prix	45		Produit	Prix	Qté			
3				Chemise	30	200			
4			Ligne 2	Jupe	45	300			
5				Pantalon	55	150			
6					Colonne 1				

Premier argument	Deuxième argument	Troisième argument
\$E\$3:\$F\$5	EQUIV(B1;\$D\$3:\$D\$5;1)	EQUIV(A2;\$E\$2:\$F\$2;1)
Table contenant les données	Automatise la détermination de la ligne	Automatise la détermination de la colonne
Les listes étant triées, le troisième argument des fonctions EQUIV peut être soit omis, soit égal à 1		

2.4 Imbrication des fonctions

Les fonctions RECHERCHEV Ou H, INDEX et EQUIV peuvent s'imbriquer et se compléter.

Une imbrication de RECHERCHEV est construite comme suit :

=RECHERCHEV(Valeur_cherchée ;Table_matrice ;Numéro_index_colonne ;valeur_proche(VRAI/FAUX))


Recherche Externe	Trouvée par une recherche interne	Arguments 2,3 et 4 de la recherche Externe
-------------------	--	--

RECHERCHEV(Valeur_cherchée ;Table ;Numéro_index_colonne ;VRAI/FAUX)

- Le résultat de la fonction RECHERCHEV la plus interne devient le premier argument de la fonction RECHERCHEV externe. Celle-ci détermine donc une valeur de ce premier argument
- La fonction RECHERCHEV externe effectue alors une seconde recherche dans une autre table à partir de la valeur trouvée par la recherche interne
- On peut également imbriquer plusieurs fonctions RECHERCHEV mais aussi des RECHERCHE, RECHERCHEH, EQUIV, INDEX ...

Exemple :

Détermination de la rémunération horaire en euros dans une feuille « intervention » :



La Recherche est effectuée à partir des tables PERS (feuille Personnel) et QUALIF (feuille Qualification) :

	A	B	C	D
1	Personnel intervenant			
2	Numéro	Nom	Prénom	Code Qualif
3	PERS1	JENSEN	Myriam	AC
4	PERS2	LOFOTEN	Caroline	DC
14	Zone A3:D13 nommée PERS			

	A	B	C
1	Qualification du personnel		
2	CODE QUALIF	LIBELLE	RENUMERATION HORAIRE
3	AC	Agent Commercial	27,00 €
4	DC	Directeur Commercial	62,00 €
13			
14	Zone A3:C12 nommée QUALIF		

Dans la cellule B3 de la feuille Intervention, on implante la formule suivante :

=RECHERCHEV(RECHERCHEV(A3;pers;EQUIV(Personnel!\$D\$2;Personnel!\$A\$2:\$D\$2;0));QUALIF;EQUIV(Qualification!\$C\$2;Qualification!\$A\$2:\$C\$2;0);FAUX)

La recherche interne

RECHERCHEV(RECHERCHEV(A3;pers;EQUIV(Personnel!\$D\$2;Personnel!\$A\$2:\$D\$2;0)) donne le code de la qualification (Code Qualif) du personnel dont le numéro est PERS1 (implanté en A3) soit AC.

La recherche externe détermine alors, à partir de ce code, la rémunération correspondante.

Les fonctions EQUIV permettent de retrouver les index de colonne pertinents pour chaque RECHERCHE.

2.5 Fonctions financières

Parmi les fonctions financières les plus courantes, retenons notamment :

- VA qui calcule la valeur actuelle
- TAUX qui calcule le taux d'intérêt
- NPM qui calcule le nombre de périodes
- VPM qui calcule le remboursement périodique d'un prêt.

Exemple :

Trois des quatre valeurs en B1, B2, B3 ou B4 doivent être connues. Nous précisons à chaque fois la formule à utiliser pour déterminer la quatrième valeur

	A	B	
1	Capital emprunté	20 000,00 €	$B1=VA(B2/12 ;B3 ;B4 ;0)$
2	Taux annuel	3,60%	$B2=TAUX(B3 ;B4 ;B1 ;0)*12$
3	Nombre de mois	48	$B3=NPM(B2/12 ;B4 ;B1)$
4	Mensualité	-448,01 €	$B4=VPM(B2/12 ;B3 ;B1 ;0 ;0)$

Les fonctions VAN (Valeur Actuelle Nette) et TRI (Taux de Rentabilité Interne) permettent d'évaluer les investissements à partir d'une succession de flux de trésorerie.

	A	B	C	D	E	F
1	Investissement	Flux 1	Flux 2	Flux 3		Taux
2	-10000	4000	4500	5000		8,00%
3	VAN	1531	$B3=VAN(F2;B2:D2)+A2$			
4	TRI	16%	$B4=TRI(A2:D2)$			

$VAN(TAUX ;VALEUR1 ; VALEUR2 ; ...)$

$TRI(VALEUR1 ; VALEUR2 ;...)$

2.6 Fonctions statistiques

Les fonctions dites statistique sont nombreuses :

Les principales

Fonctions	Exemple et résultat
ECARTYPE	=ECARTYPE(C5 :C28) Affiche l'écart type des valeurs de la plage C5 :C28
PENTE	=PENTE(A2 :F2 ;A1 :F1) Donne la pente (a) d'une droite d'ajustement d'équation $y = ax + b$ A2 :F2 → Valeurs y et A1 :F1 → Valeur x
TENDANCE	=TENDANCE(A2 :F2 ;A1 :F1 ;g1) Effectue une extrapolation linéaire et donne la valeur de y devant correspondre à la valeur de x contenue dans G1
COEFFICIENTCORRELATION	Donne le coefficient de corrélation de deux séries de valeurs
ORDONNEE.ORIGINE	Donne l'ordonnée d'origine (b) de la droite $y = ax + b$
DROITEG(utilisée de façon matricielle)	Donne la pente et l'ordonnée d'origine d'une droite d'ajustement

Figure 4 Principales fonctions statistiques

2.7 Ergonomie d'une feuille de calcul

2.7.1 Fonction d'information EST

Les fonctions EST sont utiles pour tester le contenu de cellules. Elles ne contiennent qu'un seul argument, obligatoire, qui est une valeur à tester.

L'argument valeur peut-être :

- Une valeur vide, c'est-à-dire une cellule qui ne contient aucune donnée ou formule et non qui n'affiche rien
- Une erreur renvoyée par le tableur
- Une valeur logique, de texte, de nombre ou une valeur de référence ou un nom s'y référant.

Quand la fonction EST n'est pas associée à une autre fonction elle renvoie VRAI ou FAUX.

Les fonctions d'information les plus courantes sont au nombre de sept :

ESTVIDE	Renvoie vrai si l'argument valeur est vide
ESTERREUR	Renvoie vrai si l'argument valeur fait référence à une valeur d'erreur
ESTLOGIQUE	Renvoie vrai si l'argument valeur fait référence à une valeur logique
ESTNA	Renvoie vrai si l'argument valeur fait référence à une valeur d'erreur #N/A
ESTNUM	Renvoie vrai si l'argument valeur représente un nombre
ESTREF	Renvoie vrai si l'argument valeur est une référence
ESTTEXTE	Renvoie vrai si l'argument valeur se présente sous forme de texte

Figure 5 Utilité des principales fonctions EST

D'autres fonctions d'information sont accessibles et documentées dans l'aide du tableur.

Exemple :

L'amicale laïque utilise un tableur pour gérer ses salles de réunion.

	A	B	C	D
	Catégorie Salle	Capacité d'Accueil	Capacité connue VRAI/FAUX	Capacité inconnue VRAI/FAUX
1				
2	Secrétariat	Inconnu	FAUX	VRAI
3	Yoga	23	VRAI	FAUX
4	Arts plastiques	30	VRAI	FAUX
5	Comptabilité	inconnu	FAUX	VRAI

=ESTNUM(B2), implanté en C2, affiche VRAI si C2 contient une valeur numérique, FAUX, dans le cas contraire.

B2 contient du texte (inconnu) donc la fonction renvoie FAUX

A l'inverse, =ESTTEXTE(B2) renvoie VRAI si le contenu est du texte et donc inconnu et FAUX dans le cas contraire.

Les fonctions EST sont fréquemment associées à une alternative FONCTION SI pour renvoyer un message pertinent à l'utilisateur du tableur en cas d'erreur, d'absence de saisie et/ou pour effectuer un calcul sous condition.

Exemple :

Le modèle suivant a été implanté dans le tableur pour automatiser la création des factures de l'entreprise Rouillard. Ce modèle fait notamment référence à une table produit, les informations étant récupérées par des formules utilisant la fonction RECHERCHEV.

	A	B	C	D	E	F
1	FACTURE					
2	Numéro		Date		20%	
3	Référence	Désignation	quantité	HT	Total TTC	
4	SNO	Snow	1	147,85	177,42	
5	TIR	#N/A	12	#N/A	#N/A	
6		#N/A		#N/A	#N/A	

	A	B	C
1	Référence	Désignation	PUHT
2	ARY	Arys	7,20
3	BAR	Barathéon	27,41
4	CAS	Castral roc	201,25
5	CAT	Cateylen	85,36
6	SNO	Snow	147,85
7	STA	Stark	8,36
8	TYR	Tyrion	120,36
9	VAR	Varys	258,65
10	VEN	Vent gris	25,14
11	WIN	Winterfell	29,50
12			
13	Plage A2:C11 nommée PRODUITS		

De nombreuses erreurs apparaissent. Pour pallier le caractère peu professionnel du document, il est nécessaire de recourir aux fonctions d'informations :

	A	B	C	D	E	F	G	H	I	J	K	L
1	FACTURE											
2	Numéro		Date		20%							
3	Référence	Désignation	quantité	HT	Total TTC							
4	ARY	Arys	1	7,2	8,64							
5	TIR	Réf Erronée	12	Réf Erronée								

Les formules qui permettent de gérer les erreurs ou absences de saisies sont les suivantes :

B4=SI(ESTVIDE(A4);"";SI(ESTNA(RECHERCHEV(A4;PRODUITS;2;FAUX));"Réf Erronée";RECHERCHEV(A4;PRODUITS;2;FAUX)))

D4=SI(ESTVIDE(A4);"";SI(ESTNA(RECHERCHEV(A4;PRODUITS;3;FAUX));"Réf Erronée";RECHERCHEV(A4;PRODUITS;3;FAUX)))

E4=SI(ESTNUM(C4*D4);C4*D4;""))

2.7.2 Fonction SIERREUR

La fonction SIERREUR permet également de mettre en évidence les erreurs de calcul.

C'est une fonction logique, sa portée est plus générale et son usage plus simple que les fonctions d'information portant sur des erreurs car elle permet de traiter toutes les erreurs renvoyées par le tableur.

Sa syntaxe est la suivante :

=SIERREUR(valeur_testée ;valeur_si_erreur)

La valeur testée peut être une formule, une référence, une donnée ; elle est renvoyée s'il n'y a pas d'erreur. L'argument « valeur_si_erreur » peut prendre la forme d'un message affiché alertant l'utilisateur sur le type d'erreur.

Exemple :

La formule implantée en B3 dans le cas de l'entreprise Rouillard devient :

B4=SI(ESTVIDE(A4);"";SIERREUR(RECHERCHEV(A4;PRODUITS;2;FAUX);"Réf Erronée"))

2.7.3 Fonction CONCAT

La fonction CONCAT (anciennement CONTATENER) est une fonction de texte qui permet de regrouper plusieurs valeurs en une seule.

Définition :

Une **valeur** est un texte entre guillemets ou une formule ou une référence à une cellule contenant une formule.

Cette fonction permet de combiner, dans une seule chaîne de caractères, le résultat d'une formule et un texte explicatif. Elle est donc utile en gestion.

Sa syntaxe est la suivante :

=CONCAT(valeur1 ; valeur2 ; ...)

Exemple :

La formule implantée dans le cas Gallo en A6 permet de personnaliser par une phrase l'affichage du résultat des calculs pour le client dont le nom est en B1.

=CONCAT("Le montant de la ristourne du client ";B1;" est de ";B4;" euros, pour un taux de ";RECHERCHEH(B2;E3:J4;2;VRAI))													
	A	B	C	D	E	F	G	H	I	J	K	L	
1	Nom client	GALLO		Pourcentages de ristournes octroyées selon le montant total des chiffres d'affaires (intervalles de ces montants en colonnes ci-dessous)									
2	Chiffre d'affaires	125 000,00 €	Sens de la recherche										
3	Taux de ristourne	0,92%		Seuils	0	50 000	10 000	200 000	400 000	600 000			
4	Montant de ristourne	1 150,00 €	Taux	0,75%	0,80%	0,92%	0,96%	0,97%	0,99%				
5				Ligne 2 : données correspondantes									
6	Le montant de la ristourne du client GALLO est de 1150 euros, pour un taux de 0,0092												

A6 =CONCAT("Le montant de la ristourne du client ";B1;" est de ";B4;" euros, pour un taux de ";RECHERCHEH(B2;E3:J4;2;VRAI))

L'opération de **concaténation** peut contenir au maximum 255 éléments et un total de 8192 caractères.

2.7.4 Formats de cellules

Excel offre de nombreuses possibilités pour adapter le format des cellules aux documents de gestion à produire en fonction du contexte.

Le menu ACCUEIL regroupe l'essentiel des possibilités offertes :

- Polices de caractères (type, style, taille, couleur, soulignement...)
- Alignement (Orientation, positionnement, renvoi à la ligne, fusionner les cellules ...)
- Nombre (Standard, monétaire, comptabilité, date personnalisée ...)
- Bordure (Affichage, remplissage, personnalisé ...)
- Style (Affichage conditionnel, tableaux prédéfinis, styles prédéfinis ...)
- Cellules (Hauteur ligne, largeur colonne, couleur des onglets)
- Protection et tris

Exemple :

Voici les règles de formatage conditionnel des notes obtenues au DCG.

Gestionnaire des règles de mise en forme conditionnelle

Afficher les règles de mise en forme pour : Sélection actuelle

Nouvelle règle... Modifier la règle... Supprimer la règle

Règle (appliquée dans l'ordre indiqué)	Format	S'applique à	Interrompre si Vrai
La valeur de la cellule est comprise...	AaBbCcYyZz	= \$B\$5:\$D\$6;\$B\$8:\$D\$8;\$F\$8:\$N\$8;\$	<input type="checkbox"/>
Valeur de la cellule < 6	AaBbCcYyZz	= \$B\$5:\$D\$6;\$B\$8:\$D\$8;\$F\$8:\$N\$8;\$	<input type="checkbox"/>
Valeur de la cellule > 10	AaBbCcYyZz	= \$B\$5:\$D\$6;\$B\$8:\$D\$8;\$F\$8:\$N\$8;\$	<input type="checkbox"/>

OK Fermer Appliquer

	A	B	C	D	E	L	M	N
1	DCG RESULTATS AUX EXAMENS							
2	ANNEE	2016						
3	Coefficient	1	1	1	1	1,5	1	1
4		UE1	UE2	UE3		UE11	UE12	UE13
5	ETU 1	8	12	11,75		15	12,5	11,75
6	ETU 2	8	8,75	10,25		14,5	5	11,25
7	.../...							
8	ETU 40	10	10	10		14,5	11,25	8,25

2.7.5 Validation des données

Il est possible d'ajouter des **règles de validation** des données pour une cellule ou une plage de cellules. Elles sont regroupées dans le menu « DONNEES /Validation des données ».

On peut spécifier le type de données saisies (décimales, liste de valeurs autorisées, dates, pour un intervalle de validation...).

Un message personnalisé peut être associé à la saisie.

Des messages d'alerte peuvent également être sélectionnés.

2.8 SYNTHÈSE

Le tableur : Fonctions de recherche, contrôle et ergonomie des feuilles

Les fonctions de recherche

Forme générale	=RECHERCHE(valeur à chercher ;liste où l'on recherche ;liste de résultats)
Recherche verticale par colonnes	=RECHERCHEV(Valeur_cherchée ;Table_matrice ;numéro_index_colonne ;valeur_proche (VRAI/FAUX))
Recherche horizontale par lignes	=RECHERCHEH(Valeur_cherchée ;Table_matrice ;numéro_index_ligne ;valeur_proche (VRAI/FAUX))
Recherche à l'intersection d'une ligne et d'une colonne	=INDEX(Tableau ;no_ligne ;no_colonne)

Les fonctions de recherche spécifiques

Les numéros des colonnes ou des lignes peuvent être déterminés par la fonction EQUIV :

=EQUIV(valeur_cherchée ;matrice_recherche ;type)

Les fonctions RECHERCHEV, RECHERCHEH, INDEX et EQUIV peuvent s'imbriquer et se compléter.

Les fonctions de test logique et la validation des données :

- Les fonctions EST sont utiles pour tester le contenu de cellules et permettre la gestion des erreurs, combinées à la fonction SI et à un message d'erreur pertinent. Elles améliorent l'ergonomie des feuilles de calcul
- La fonction logique SIERREUR joue le même rôle avec une portée plus générale
- Les formats doivent être adaptés au contexte et répondre à des exigences de qualité du document qui doit jouer un rôle professionnel.
- La validation des données permet également d'améliorer l'ergonomie et de limiter les erreurs de saisie